
The goal of this activity is to modify a program written in a legacy version of 
Fortran (F77), to incorporate aspects of modern Fortran, remove redundancies, 

and obsolete structures, i.e. get rid of all the GOTOs and find some sort of 
structure which can be used to make the program more readable, and less 

unstructured.

SYNOPSIS
The goal of this activity is to modify a program written in a legacy version of 

Fortran (F77), to incorporate aspects of modern Fortran, remove redundancies, 
and obsolete structures, i.e. get rid of all the GOTOs and find some sort of 
structure which can be used to make the program more readable, and less 

unstructured.

Type: legacy re-engineering

Language: Fortran

Compiler: gfortran

Skills: Fortran re-engineering, re-writing legacy programs, 
language skills, removing goto’s

Experience Level: intermediate

“By systematic comparisons of the patterns of logic, at million-per-second speeds, 
computers can explore the realms of the intellect in details never before perceived.”

Donald D. Fink.

PROGRAM DESCRIPTION

The program below converts a year, and a day of  the year (in Julian code), into the month and 
the day of  the month. The excessive use of  GOTOs and labels makes such code reek of  
unstructured spaghetti-code, and its associated undisciplined design principles. 

THE CODE

C      DAY OF YEAR
       PROGRAM DAY_OF_YEAR
5      READ (*,*) IYEAR, IDATE
       IF (IDATE .LT. 1 .OR. IDATE .GT. 366 .OR. IYEAR .LT. 0) GO TO 999
       IF (IDATE .LE. 31) GO TO 800 
       L = 1
       I = IYEAR/400; IF (IYEAR-(I*400) .LE. ZERO)  GO TO 20 
       I = IYEAR/100; IF (IYEAR-(I*100) .LE. ZERO) GO TO 10 
       I = IYEAR/4; IF (IYEAR-(I*4) .LE. ZERO) GO TO 20 
10     L = 0
       IF (IDATE .GT. 365)  GO TO 999
20     IF (IDATE .GT. 181 + L) GO TO 181
       IF (IDATE .GT. 90 + L) GO TO 90 
       IF (IDATE .GT. 59 + L) GO TO 59 
       MONTH = 2; IDAY = IDATE - 31; GO TO 900
59     MONTH = 3; IDAY = IDATE - (59 + L); GO TO 900
90     IF (IDATE .GT. 120 + L) GO TO 120
       MONTH = 4; IDAY = IDATE - (90 + L); GO TO 900
120    IF (IDATE .GT. 151 + L) GO TO 151
       MONTH = 5; IDAY = IDATE - (120 + L); GO TO 900
151    MONTH = 6; IDAY = IDATE - (151 + L); GO TO 900
181    IF (IDATE .GT. 273 + L) GO TO 273
       IF (IDATE .GT. 243 + L) GO TO 243
       IF (IDATE .GT. 212 + L) GO TO 212
       MONTH = 7; IDAY = IDATE - (181 + L); GO TO 900
212    MONTH = 8; IDAY = IDATE - (212 + L); GO TO 900
243    MONTH = 9; IDAY = IDATE - (243 + L); GO TO 900
273    IF (IDATE .GT. 334 + L) GO TO 334
       IF (IDATE .GT. 304 + L) GO TO 304
       MONTH = 10; IDAY = IDATE - (273 + L); GO TO 900
304    MONTH = 11; IDAY = IDATE - (304 + L); GO TO 900
334    MONTH = 12; IDAY = IDATE - (334 + L); GO TO 900
800    MONTH = 1
       IDAY = IDATE
       GO TO 900 
900    WRITE (*,*) MONTH, IDAY, IYEAR
       GO TO 5
999    WRITE (*,*) "error with data? ", IDATE, IYEAR 
       END

See listing doy_legacy.for

PROGRAM ANALYSIS

So the program runs. Great. I mean its only 39 lines long - how difficult could it be to remove a 
few GO TO statements? The thing is it’s not only an abundance of  unstructured branches that 
have to be dealt with. It’s also the lack of  usability. Run the program, and it prompts for nothing, 
assuming you know what it needs - and indeed can decipher the output. Here is an example of  
the programming running:

2000 64
           3           4        2000
1900 12
           1          12        1900
^C

As input it wants the year, and number representing the day of  the year, short of  looking at the 
code, nobody would know that. To stop the program you have to use CTRL-C. 

There are 39 lines of  code, and 29 statements involving GO TO. Of  course it is kind-of  
deceptive - many of  the lines contain multiple statements to make the code seem more compact 
than it would otherwise. The code:

	 MONTH = 7; IDAY = IDATE - (181 + L); GO TO 900

is actually 3 statements, and could just as well be written as:

	 MONTH = 7
! IDAY = IDATE - (181 + L)
! GO TO 900

Twelve of  the GO TO statements actually branch to label 900 - where the calculated month and 
day are printed out. There must be a better way of  writing this program - and there is, through 
the process of  re-engineering it.

Oh, did I mention there is no documentation? No really, none. That’s not uncommon, and it will 
lead to a review of  what’s happening in the program. 

When looking through the program, the first thing processed after receiving the user input is to 
determine whether the year input is a leap year. This sets a variable to hold a value 1 or 0, based 
on whether it is a leap year, or not. Next the program goes on to determine where the date sits 
based on its value.

FIRST THINGS FIRST

The first thing to do is change the file extension to .f95, and modify the comment delimiter from 
C to !. The file will still compile and run.

See listing doy.f95

IMPROVE READABILITY

To improve readability the first task is to convert the program from uppercase to lowercase. This 
can be achieved using the “translate characters” utility, tr:

! tr '[:upper:]' '[:lower:]' < doy.f95 > doy_1.f95

Note that at the moment, the program is not format-free, this will be done after the jump labels 
have been removed, so there is some clarity during the re-engineering process. It can be indented 
as the re-engineering progresses to improve program clarity. Each major update will be provided 
in a separate file to illustrate the evolution of  the re-engineering process.

Now to make things easier to read, reorganize the file so that each statement is on a line of  its 
own. The program is now 64 lines long.

See listing doy_1.f95

FIRST FIX THE EASY STUFF

Of  the 29 statements involving GO TO, 13 of  them are unconditional GO TO’s, and 16 are 
associated with jumping from an if statement. This is where things *can* get messy. Best to deal 
with the simple stuff  first - but that requires reading through the program and trying to 
comprehend what it does. Don’t panic. Start at the top.

The first real code encountered deals with determining whether or not the input is valid.

    if (idate .lt. 1 .or. idate .gt. 366 .or. iyear .lt. 0) go to 999

If  it is invalid, then it passes control to label 999, which prints an error message and terminates 
the program. The program says “So sad, too bad”. Apart from the go to statement, we can clean 
up the syntax to F95 standards. The code now becomes:

    if (idate < 1 .or. idate > 366 .or. iyear < 0) then
        go to 999
    end if

Basically change the format of  the if statement, and modify the operators. Then deal with the 
issue of  idate having a value less than or equal to 31 - signifying the month is January and the 
day equals idate. So

    if (idate .le. 31) go to 800

becomes

    if (idate <= 31) then
        go to 800
    end if

Don’t worry about the go to’s - we’ll fix them eventually. 
The next segment of  code deals with the leap year problem.

       l = 1
       i = iyear/400
       if (iyear-(i*400) .le. zero)  go to 20
       i = iyear/100
       if (iyear-(i*100) .le. zero) go to 10
       i = iyear/4
       if (iyear-(i*4) .le. zero) go to 20
10     l = 0

It looks confusing because early Fortran (<F77) did not have any easy way of  calculating integer 
remainders, i.e the mod function. So a lot of  extra code was needed to get around this. Basically 
this code checks the value of  iyear in sequence:

 Is iyear divisible by 400? Yes, branch to label 20 (l = 1, it’s a leap year!).
	 No?
 Is iyear divisible by 100? Yes, branch to label 10 (l = 0, it’s not a leap year!).
	 No?
 Is iyear divisible by 4? Yes, branch to label 20 (l = 1, it’s a leap year!).

What is needed is to have these statements consolidated into one if-else statement using the 
following logic:

! if [iyear is divisible by 400 OR 
!     ((iyear is not divisible by 100) AND 
           (iyear is divisible by 4))] then
! ! iyear is a leap year, l=1
! else
! ! iyear is not a leap year, l=0
	
This reveals itself  in the following code, using the mod function to :

    if ((mod(iyear,400)==0) .or. ((mod(iyear,100)/=0) .and. 
                                  (mod(iyear,4)==0))) then
        l = 1
        go to 20
    else
        l = 0
    end if

See listing doy_2.f95

FIX THE MULTIPLE GOTO’S

Fixing the core part of  the program is wearisome more than anything. Looking at the diagram 
outlining the progression of  jumps, there is sequential feel to how it works. Fixing the structure is 
a process of  replacing goto statements with the code at the label. Here’s an example, the original 
code looks like this:

20     if (idate > 181 + l) go to 181
       if (idate > 90 + l) go to 90
       if (idate > 59 + l) go to 59
       month = 2
       iday = idate - 31
       go to 900

The first step is to reorganize the if statements into if-then-else statements:

20     if (idate > 181 + l) then
           go to 181
       else if (idate > 90 + l) then
           go to 90
       else if (idate > 59 + l) then
           go to 59
       else
           month = 2
           iday = idate - 31
           go to 900
       end if

Then replace the goto statements with the code found at the label. When the jump to label 181 
in the code above is replaced, the new code looks like this:

20     if (idate > 181 + l) then
           if (idate > 273 + l) go to 273
           if (idate > 243 + l) go to 243
           if (idate > 212 + l) go to 212
           month = 7
           iday = idate - (181 + l)
           go to 900
       else if (idate > 90 + l) then
           go to 90
       else if (idate > 59 + l) then
           go to 59
       else
           month = 2
           iday = idate - 31
           go to 900
       end if

This code can now be reconfigured so that the if statements are again transformed into if-then-
else statements.

20     if (idate > 181 + l) then
           if (idate > 273 + l) then
               go to 273
           else if (idate > 243 + l) 
               go to 243
           else if (idate > 212 + l) 
               go to 212
           else
               month = 7
               iday = idate - (181 + l)
               go to 900
           endif
       else if (idate > 90 + l) then
           go to 90
       else if (idate > 59 + l) then
           go to 59
       else
           month = 2
           iday = idate - 31
           go to 900
       end if

This process continues until all the goto’s associated with if statements are removed. Once this 
has occurred, all the “go to 900” statements can be deleted, as they are now redundant. Once 
this has occurred, there should only be five goto statements in the code. 

TEST THE CODE - LIKE NOW!

Before removing the final goto statements, the program should be thoroughly tested (actually 
testing the program as you are rationalizing the goto/if combinations is a good idea too). The 
best way of  doing this is by feeding the program every possible day-of-year number. An easy way 
of  doing this is writing a quick program to generate an input file. Here is a simple C program:

#include <stdio.h>

int main(void)
{
    FILE *ofp;
    int i, year;

    printf("Year: ");
    scanf("%d", &year);
    ofp = fopen("days.txt","w");

    for (i=1; i<=365; i=i+1)
        fprintf(ofp,"%d %d\n",year,i);

    fclose(ofp);

    return 0;
}

It generates a file named days.txt, which contains 365 lines, each equivalent to the input need 
for the Fortran program. When run in the following manner:

! ./a.out < days.txt

it will produce output for every day for that year. This can quickly be checked to see if  there are 
any discrepancies. 

See listing doy_3.f95

THE FINAL GOTOs

The goto statement associated with label 5 which essentially runs through the algorithm 
numerous times, can be easily replaced by a do loop surrounding the whole algorithm. The error 
messages associated with the label 999 can be re-written to incorporate the error message where 
the error is trapped. For example:

    if (idate < 1 .or. idate > 366 .or. iyear < 0) then
        go to 999
    end if
    ...
    ...

    999    write (*,*) "error with data? ", idate, iyear

now becomes:

    if (idate < 1 .or. idate > 366 .or. iyear < 0) then
        write (*,*) "error with data? ", idate, iyear
        exit
    end if

The error message is printed, and the do loop is exited, effectively ending the program. Next, the 
goto and jump associated with label 20 can be removed. At present the code looks like this:

    if ((mod(iyear,400)==0) .or. ((mod(iyear,100)/=0) 
                            .and. (mod(iyear,4)==0))) then
        l = 1
        go to 20
    else
        l = 0
    end if

    if (idate > 365) then
        write (*,*) "error with data? ", idate, iyear
        exit
    end if

The goto basically “skips” over the if statement below it. If  the if statement is incorporated into 
the else above it, then there is no need for the “go to 20”. Here is the fixed code:

    if ((mod(iyear,400)==0) .or. ((mod(iyear,100)/=0) 
                            .and. (mod(iyear,4)==0))) then
        l = 1
    else
        l = 0
        if (idate > 365) then
            write (*,*) "error with data? ", idate, iyear
            exit
        end if
    end if

One final goto to deal with, which sits in this code:

    if (idate <= 31) then
        month = 1
        iday = idate
        go to 900
    end if

Fixing this involves printing out the result, and cycling to the next instance of  the loop. Here is 
the rehashed code:

    if (idate <= 31) then
        month = 1
        iday = idate
        write (*,*) month, iday, iyear
        cycle
    end if

The code is now re-engineered and works very well. However look at the code. Is this the best 
way to deal with this algorithm?

See listing doy_4.f95

A BETTER ALGORITHM?

The program as it stands seems somewhat long-winded. This is a case where the algorithm may 
not be very efficient, at least from the point of  view of  coding efficiency (not how efficient it runs). 
Is there a better way to express this algorithm? As the algorithm is expressed in “cumulative 
month days”, there should be a way of  using this data as an array and sequentially working 
through it. Consider the following algorithm:

1. Store the data in an array of  the form (the first row comprises the cumulative month days for 
a non-leap year, the second row for a leap year:

31 59 90 120 151 181 212 243 273 304 334 365

31 60 91 121 152 182 213 244 274 305 335 366

2. Determine if  the year is a leap year or not, leap = 0 or 1 - determines which row of  the array 
to use in calculations. 

3. Deal with errors in the data, i.e. year or date out of  bounds.
4. Loop through the data while the date is less than the array element. For example if  date=157,    

on a non-leap year, the loop will cycle through 0, 31, 59, 90, 120, 151. So the month becomes  
5, and the day becomes 157-151 = 6. If  the month=1, then the day is simply the date.

Here’s the code to achieve this:

program day_of_year

integer :: iyear, idate, leap, month
integer, dimension(0:1,1:12) :: ndays

ndays = reshape( (/ 31, 31, 59, 60, 90, 91, 120, 121, 151, 152, &
                    181, 182, 212, 213, 243, 244, 273, 274, 304, 305, &
                    334, 335, 365, 366 /), (/2,12/))
do
    month = 1
    read (*,*) iyear, idate
    if (mod(iyear,400)==0 .or. (mod(iyear,100)/=0 .and. 
        mod(iyear,4)==0)) then
        leap = 1
    else
        leap = 0
    end if

    if ((iyear < 1753) .or. (iyear > 3999) .or. (idate <= 0) .or. 
        (idate > ndays(leap,12))) then
        write (*,*) "Error: bad year, date"
    else
        do while (idate > ndays(leap,month))
            month = month + 1
        end do

    end if
    if (month == 1) then
        write (*,*) "month = ", month, " day = ", idate
    else
        write (*,*) "month = ", month, " day = ", idate-ndays(leap, month-1)
    end if
end do

end program day_of_year

See listing doy_better.f95

